NASA TECHNICAL MEMORANDUM

NASA TM X-52234

NASA TM X-52234

A QUASI-THREE-DIMENSIONAL METHOD FOR CALCULATING BLADE SURFACE VELOCITIES FOR AN AXIAL FLOW TURBINE BLADE

by Theodore Katsanis and Lois T. Dellner
Lewis Research Center
Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION · WASHINGTON, D.C. · 1966

A QUASI-THREE-DIMENSIONAL METHOD FOR CALCULATING BLADE SURFACE VELOCITIES FOR AN AXIAL FLOW TURBINE BLADE

by Theodore Katsanis and Lois T. Dellner

Lewis Research Center
Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

E-3580

A QUASI-THREE-DIMENSIONAL METHOD FOR CALCULATING BLADE SURFACE VELOCITIES FOR AN AXIAL FLOW TURBINE BLADE

by Theodore Katsanis and Lois T. Dellner

Lewis Research Center
National Aeronautics and Space Administration
Cleveland, Ohio

SUMMARY

The aerodynamic design of turbine stator or rotor blades requires accurate determination of the velocity distribution on the blades. For axial flow turbines a quasi-three-dimensional compressible flow analysis has been used successfully for many years. The analysis is based on two methods, one to obtain a blade-to-blade variation of velocity, and the other to obtain the radial variation in velocity. A combination of the methods gives a quasi-three-dimensional compressible flow analysis. Most of the calculations have been incorporated into a program. This report gives a description of the quasi-three-dimensional flow analysis and the FORTRAN IV computer program.

INTRODUCTION

The aerodynamic design of axial flow nozzles and turbine blades requires accurate determination of the velocity distribution on the blades. In the flow analysis, three-dimensional flow effects are of importance. Methods for accomplishing this type of analysis for blade designs of medium or high solidity have been developed at Lewis Research Center (refs. 1 to 4).

Reference 1 gives a method for determining the variation of velocity from hub to tip on a mid-channel stream surface, and reference 2 gives a method for determining the blade-to-blade variation velocity. A combination of the two methods gives a quasi-three-dimensional flow analysis. Weight flow calculations are then based on a two-dimensional integration across a passage cross section. Most of the basic calculations needed for this analysis have been incorporated into a computer program to facilitate practical turbine design. The program is referred to as CTTD (Compressor and Turbine Division Turbine Design program).

The purpose of this report is to give a complete description of the quasi-three-dimensional flow analysis and of the use of the CTTD computer program. This technique has been used at Lewis Research Center for 14 years and the program has long been available to industry for their own use in turbine design. The original work of developing the analysis method and the CTTD program was done by the staff of the Compressor and Turbine Division at NACA Lewis Flight Propulsion Laboratory.

SYMBOLS

a	parameter, eq. (A1) and (A2)
b	parameter, eq. (A1) and (A2)
C	curvature of streamlines on blade-to-blade surface, 1/ft
g	gravitational constant, ft/sec^2
m	distance along meridional streamline, ft
n	distance along orthogonal to streamline, ft
n_0	distance along orthogonal to streamline on blade-to-blade surface from suction to pressure surface, ft
p	pressure, lb/ft^2
r	radius, ft
r_c	radius of curvature of streamline in radial-axial plane, ft
R	gas constant, $(ft)(lb)/(lb)(^{\circ}R)$
T	temperature, deg R
V	absolute velocity, ft/sec
w	weight flow, lb/sec
W	relative velocity, ft/sec
z	axial coordinate, ft
α	angle of meridional streamline with the axial direction, deg.
β	angle of streamline on blade to blade surface with the axial direction, deg. (see fig. 3)
γ	specific heat ratio
ρ	density, lb/ft^3
ω	rotational speed, radians/sec

Subscripts

calc	calculated
cr	critical velocity
e	exit
giv	given
h	hub
i	inlet
m	mean
mid	mid-channel
p	pressure
s	suction
t	tip
θ	tangential direction

Superscripts

'	absolute total condition
"	relative total condition

METHOD AND ASSUMPTIONS

The objective of the analysis method is to calculate the quasi-three-dimensional velocity distribution satisfying continuity at a given channel orthogonal surface (see fig. 1). The weight flow may be specified in the calculation, or the calculation may continue until the maximum (choking) weight flow for that channel orthogonal surface is determined. The velocity variation on the orthogonal surface is calculated from radial equilibrium (ref. 1) and stream filament theory (ref. 2). When this is done for several orthogonal surfaces, a velocity distribution over the blade surface is determined. The velocity distribution can thus be obtained for the guided channel formed by the portion of the passage where the blade-to-blade orthogonals extend from suction to pressure surface. The guided channel will not cover the entire suction surface. To obtain a velocity on the uncovered portion of suction surface some method must be used to estimate the location of the pertinent stagnation streamline. Since these velocities depend critically on the location of the stagnation streamline, unreliable results may be obtained on the uncovered portion of the blade.

The basic simplifying assumptions used in deriving the equations used are:

1. The flow relative to the blade is steady.

2. The fluid is a perfect gas.

3. The fluid is a nonviscous gas.

4. The fluid velocity has no radial component. (The projections of the streamlines on a radial-axial plane are straight lines parallel to the axis.)

5. The mid-channel line is a streamline, hereinafter referred to as the mid-channel streamline.

6. (a) The gas has a constant entropy from hub to tip at the mid-channel streamline at a fixed axial coordinate. This assumption is used in calculating the mid-channel velocity variation (ref. 1).

(b) The gas has a constant entropy blade-to-blade along an orthogonal to the streamlines. This assumption is used in calculating the blade-to-blade velocity variation (ref. 2).

7. A line connecting the midpoints of the hub, mean, and tip orthogonals at a fixed axial location in the channel is a straight radial line. This assumption is used in calculating the mid-channel velocity variation (ref. 1).

8. The meridional streamline curvature varies linearly from hub to tip. (By assumption 4, meridional streamline curvature would be zero; however, the effect of wall curvature on radial equilibrium may be considered.)

9. There is free vortex velocity distribution at the inlet to the blade; that is, $(rV_\theta)_i$ is a constant from hub to tip. This assumption is used in calculating the mid-channel velocity variation (ref. 1).

10. Uniform inlet absolute total temperature, T_i'. This assumption is used in calculating the mid-channel velocity variation (ref. 1).

11. An additional assumption is necessary to calculate the velocity variation from blade-to-blade along an orthogonal. An option is provided for this assumption in the program. The usual assumption is that there is a linear variation of streamline <u>curvature</u> along an orthogonal. An alternate assumption is that there is a linear variation of <u>radius of curvature</u> along an orthogonal. There is no particular reason why one assumption is preferred over the other. For high solidity blading, it makes very little difference which assumption is chosen.

Another possibility is provided in the program. In this case the blade surface velocities are calculated as above based on the assumption of linear variation of either curvature or radius of curvature. Then for the weight flow calculation, density and velocity along the orthogonal are computed by assuming a linear variation of static pressure along the orthogonal. This is not consistent with the assumption of linear variation of either curvature or radius of curvature and should be used with caution.

FLOW CHANNEL LAYOUT AND ENGINEERING DATA

The initial steps in the design, those involving the development of the inlet and outlet velocity diagrams, will not be described (see ref. 5 for further information). It is assumed that these velocity diagrams have been obtained, as well as the basic operating conditions of design weight flow, number of blades, gas to be used, operating speed and inlet and outlet stagnation temperature and pressure. From this information an initial blade shape can be drawn (ref. 6). The following steps are followed to obtain a blade surface velocity distribution.

A. Along a section midway between the hub and tip, a cylindrical development of the blade channel should be accurately laid out to scale several times actual size. Figure 2 shows a typical blade channel - mean section. (This section need not be at constant radius, but the meridional streamline angle, α, should be small enough that $\cos \alpha$ is approximately 1.) The mid-channel streamline is then drawn midway between the suction and pressure surfaces, as shown in figure 2. This procedure is repeated for the hub and tip of the blade. The axial coordinate locations of the hub, mean, and tip sections relative to each other must be established. For this, an axial coordinate reference line is specified on each of the three blade sections. The relative angular location, θ, of the hub, mean, and tip sections is not considered; the effect is usually neglible.

B. Any number of axial locations can now be chosen on the three mid-channel streamlines. The distance along the mid-channel streamline between each axial location should be measured for use in specifying loss distribution. At any given axial location a curve is drawn through the mid-channel streamline from blade-to-blade so as to be orthogonal to each blade surface and to the mid-channel streamline, as illustrated in figure 3. This is done at hub, mean, and tip. The corresponding orthogonals at the three radial blade stations are positioned so that the intersections of the mid-channel streamlines and the orthogonals are located at the same axial coordinate. In general, the intersections of the orthogonals with the suction and pressure surfaces at the three radial sections will not be at the same axial coordinates. These three orthogonals determine a section through the blade passage over which the velocity variation will be determined and across which weight flow will be calculated. The total length of the orthogonal, n_O, is required as input for the program. Also the blade surface curvatures, C_s and C_p, must be measured at the ends of the orthogonals. These curvatures should be measured very carefully. The angle β is the angle that the mid-channel streamline makes with the axis, and is considered positive in the direction of rotation, i.e., β is positive if the tangential component of the velocity is in the direction of rotation.

If the inner and outer walls are not straight, the inner and outer wall curvature, $1/r_c$, should be measured at each station. This curvature is considered positive if the wall is concave upwards (fig. 4). This curvature is assumed to vary linearly between hub and tip. The hub and tip radii complete the geometrical data required at each station.

C. Operating conditions must be specified. These include the gas specific heat ratio, γ, operating speed, ω, and design weight flow per channel, w. Also required are W_{cr} and ρ'' at the hub, mean, and tip for each orthogonal. The inlet relative critical velocity, W_{cr}, can be calculated using $V_{\theta,i}$ from the inlet velocity diagram together with the inlet absolute stagnation temperature T_i'.

$$W_{cr} = \sqrt{\frac{2\gamma Rg}{\gamma + 1} T''} \qquad (1)$$

where

$$T'' = T_i' - \frac{2\omega r_i V_{\theta,i} - (\omega r)^2}{2\gamma Rg} (\gamma - 1) \qquad (2)$$

By assumptions 9 and 10, T_i' and $r_i V_{\theta,i}$ are independent of radius, and equations (1) and (2) can be used to determine W_{cr} at any point in the passage as a function of radius alone.

The inlet relative total stagnation pressure can be calculated from

$$p_i'' = p_i' \left(\frac{T''}{T_i'}\right)^{\gamma/(\gamma-1)} \qquad (3)$$

The exit relative total pressure, p_e'', can be found in the same manner utilizing the exit velocity diagrams. To allow for losses, the difference, $p_i'' - p_e''$, is distributed along the length of the mid-channel streamline. Usually a linear variation is used. Finally, ρ'' is given by

$$\rho'' = \frac{p''}{RT''} \qquad (4)$$

D. The information determined above in steps B and C is the information for items 12 to 42 and 44 on the Input Data Sheet (fig. 5). One sheet is required for each axial station. The calculation at any one station is independent of any other. It will be noted that the input sheet is designed so that only the numerical values which change from sheet to sheet need to be supplied.

The output from the program includes the blade surface velocities at each orthogonal and the corresponding weight flow. These velocity calculations are made for the initial estimated value of W/W_{cr} at the mid-channel streamline for the mean blade section. It is possible for the program to determine the velocities corresponding to any desired weight flow (less than choking) or to determine the choking weight flow. If the velocity distribution corresponding to the design weight flow or some other operating weight flow is not satisfactory, the blade design should be altered, and steps A to C repeated to determine the new velocity distribution

DESCRIPTION OF PROGRAM INPUT AND OUTPUT FOR SAMPLE PROBLEM

The Input Data Sheet is shown in figure 5. The quantities filled in are for a sample problem. The output for this sample problem will be presented subsequently.

CTTD

General Instructions for Filling Out Input Data Sheets

One page must be filled out for each z(axial) station. The basic calculation at each station is from a value of $(W/W_{cr})_{mid,m}$ (hereinafter called x) to the corresponding weight flow (w_{calc}) and velocity distributions.

1 and 4 These are for problem identification and will be printed as the heading of all z stations. Thirty-nine characters are allowed on each line, and they may be alphabetic, numeric, special symbols, or blanks. It is customary to fill out these two lines on the first page only and write OMIT on these two lines on all successive pages. However, if a new heading is desired, a page which is blank except for the value "9" entered in 8 = KR4 must precede the page on which the new 1 and 4 occur.

5 JX = $\begin{pmatrix} 1 \\ 2 \end{pmatrix}$ if linear variation in $\begin{pmatrix} \text{curvature} \\ \text{radius of curvature} \end{pmatrix}$ is to be assumed in the calculation of the velocity distributions across the orthogonals.

6 JY = $\begin{pmatrix} \text{same as JX} \\ 3 \end{pmatrix}$ if linear variation in $\begin{pmatrix} \text{same as JX} \\ \text{static pressure} \end{pmatrix}$ is to be used in the calculation of the weight flow.

7 JZ = 1 if only the calculation for the one value of 44 = x is wanted.

In this case, 14 = w_{giv} is not used.

JZ = 2 if choke conditions are wanted. (Solve for the value of x which gives maximum value of w_{calc}.) In this case w_{giv} is not used.

If there is any value of x for which 14 = w_{giv} is equal to w_{calc}, there are always two such values.

JZ = 3 if the lesser of these two solutions is wanted. (subsonic)

JZ = 4 if the greater of these two solutions is wanted. (supersonic)

8 12 γ, ratio of specific heats

13 ω, wheel speed, (radians per second)

14 w_{giv}, weight flow, PER BLADE, (#/sec)

First page.

Fill out 12 → 14 and 8 = KR4 = 1.

On any successive page, if values to be entered in 12 → 14 $\begin{pmatrix}\text{differ from} \\ \text{are identical to}\end{pmatrix}$ those in 12 → 14 of the immediately preceding page, KR4 = $\begin{pmatrix}1 \\ 0\end{pmatrix}$. If KR4 = 0, write OMIT in 12 → 14.

8 = KR4 = 9 may be entered on an otherwise blank page. See explanation under 1 and 4.

9 15 $\left(1/r_c\right)_{h}$, reciprocal of radius of curvature at hub, (ft^{-1})

16 $\left(1/r_c\right)_{t}$, reciprocal of radius of curvature at tip, (ft^{-1})

17 r_h, radius at hub, (ft)

18 r_t, radius at tip, (ft)

First page.

Fill out 15 → 18 and 9 = KR5 = 1.

On any successive page, if values to be entered in 15 → 18 $\begin{pmatrix}\text{differ from} \\ \text{are identical to}\end{pmatrix}$ those in 15 → 18 on the immediately preceding page, KR5 = $\begin{pmatrix}1 \\ 0\end{pmatrix}$. If KR5 = 0, write OMIT in 15 → 18.

10 19, 24, 29 n_0, length of the orthogonal between suction and pressure surfaces at hub, mean, tip, (ft)

20, 25, 30 C_s, curvature at intersection of orthogonal with suction surface at hub, mean, tip, (ft^{-1})

21, 26, 31 C_p, curvature at intersection of orthogonal with pressure surface at hub, mean, tip, (ft^{-1})

} both must be positive (non-zero) if JX = 2

22, 27, 32 W_{cr} relative critical velocity at hub, mean, tip, (ft/sec)

23, 28, 33 $\rho''W_{cr}$ weight flow parameter at hub, mean, tip $(\#/ft^2 sec)$

First page.

Fill in 19 → 23.
If values for 24 → 28 and 29 → 33 repeat those of 19 → 23, write OMIT in 24 → 33 and 10 = KR6 = 1. If values differ, enter them and 10 = KR6 = 3.

On any successive page, if all values to be entered in 19 → 33 are identical to those in 19 → 33 of the immediately preceding page, write OMIT in 19 → 33 and KR6 = 0. If not, follow instructions given above for the first page.

11 34 → 42 β, relative (rotor) or absolute (stator) flow angle measured from the axis, (degrees), positive in the direction of rotation, mid-channel only.

First page.

If β is constant from hub to tip, fill in 34 only, write OMIT in 35 → 42 and 11 = KR7 = 1.

If β is specified only at hub, mean and tip, fill in 34, 35 and 36, write OMIT in 37 → 42 and 11 = KR7 = 3.

If β variation from hub to tip is specified (nine values), fill in 34 → 42 and 11 = KR7 = 9.

On any successive page, if all values in 34 → 42 are identical to those of the preceding page, write OMIT in 34 → 42 and 11 = KR7 = 0. If not, follow instructions given above for first page.

43 An eight digit numerical code to identify this page uniquely is entered here and will be printed out preceding the output for this z station. (Can be used to code for engineer, case and z-station number.)

44 $\left(\frac{W}{W_{cr}}\right)_{mid,m}$, critical velocity ratio, (referred to as x immediately below).

Computation of w_{calc} (and the velocity distributions) for this value of x will be performed first. If:

7 = JZ = 1, calculation stops, and input data for next station is read.

7 = JZ = 2, x is increased automatically and calculations are continued until choking weight flow has been found.

7 = JZ = 3, x is modified until subsonic solution is found. (Point at which $w_{calc} = w_{giv}$). Guess as close as possible.

7 = JZ = 4, x is modified until supersonic solution is found. Guess LOW.

Description of Computer Output

An example of the output obtained is given in table I. The first output is a listing of the information on the Input Data Sheet. In the sample output these are identified with the numbers corresponding to the numbers on the Input Data Sheet. If nine values of β (equally spaced from hub to tip) are not supplied, the missing values are calculated and printed with the input values. The remaining output is the calculated quantities. The first line is W_s/W_{mid} and W_p/W_{mid}, each at hub, mean, and tip. Next are the values of the parameters a and b of equation (A2), which are labelled LITTLE A and LITTLE B. IK = 1 is the hub and IK = 9 is the tip. ALPHA is equal to $a(r_t - r_h)/16$. M(IK) is the numerical approximation to $\int_{r_m}^{r} a(\zeta)d\zeta$ and N(IK) is $be^{-M(IK)}$. PH, PM, and PT are the numerical approximations to $\int_{r_m}^{r} b(\xi)e^{-\int_{r_m}^{\xi} a(\zeta)d\zeta} d\xi$ at the hub, mean, and tip, respectively. QH, QM, and QT are the numerical approximations to $e^{\int_{r_m}^{r} a(\zeta)d\zeta}$. The relative velocities ratioed to the relative critical velocity at the hub, mean, and tip, for suction surface, mid-channel and pressure surface are followed by the values of A(I,K) $(\rho W/\rho'' W_{cr})$. The first line is at the hub at 8 equal intervals from the suction surface to the pressure surface. The second and third line of A(IK) are at the mean and tip, respectively. On the next line N is the number of the iteration for this orthogonal, X is the value of $(W/W_{cr})_{mid,m}$ for this iteration, WT FLOW CALC is the corresponding calculated weight flow, w_{calc}, and D is the measure of error $1 - w_{calc}/w_{giv}$. If more than one iteration is done, the values down to PH, PM, PT, etc., do not change, hence for the second and following iterations only the lines following this are printed. A maximum of 5 iterations is performed. If the convergence criterion is not satisfied, the statement "NLIMIT (5) HAS BEEN REACHED" will be printed.

The message CALCULATION OF A(IK) AT ST. NO. 152+5 IN CTTD MP ASKS FOR LOG OF NEG. REMAINING CALCULATIONS THIS ITERATION THEREFORE INVALID. is usually caused by an error in the input cards. If this is not the case, the blade design must be changed.

CTTD COMPUTER PROGRAM

The program consists of the main program and two subroutines, PABC and VSUBX. PABC calculates the coefficients A, B, and C of the parabola $y = Ax^2 + Bx + C$ passing through three given points. VSUBX calculates the ratio of the velocity at any point on an orthogonal to the mid-channel velocity, using equation (A4) or (A5).

```
C    CTTD MAIN PROGRAM
      DIMENSION ENO(3),CS(3),CP(3),WCR(3),PWCR(3),VS(3),VP(3),WM(3),
     1S(3),BETA(9),ALF(9),EM(9),CAPN(9),A(9),B(9),WS(3),WP(3)
  001 EQUIVALENCE(WS(1),WSH),(WS(2),WSM),(WS(3),WST),(WP(1),WPH),(WP(2
     1),WPM),(WP(3),WPT)
       DATA NLIM,TLIM/5,.0010/
  400 FORMAT (40H1                                                      )
  401 FORMAT (40H                                                  )
  402 FORMAT (7I1)
  404 FORMAT (3E9.5)
  406 FORMAT (4E9.5)
  408 FORMAT (5E9.5)
  410 FORMAT (1H0,2I4,E9.5)
  412 FORMAT (11H0  GAMMA = ,E12.5,11H   OMEGA = ,E12.5,12H   WGIVEN = ,
     1E12.5)
  414 FORMAT (106H0                  N ZERO           C SUB S         C SUB P
     1    W SUB CR            RHO W CR       1/R SUB C              R)
  416 FORMAT (10H0   HUB     ,7E15.5)
  417 FORMAT (10H    MEAN    ,5E15.5)
  418 FORMAT (10H    TIP     ,7E15.5)
  420 FORMAT (17H0                BETA)
  422 FORMAT (1H0,I7,I4,E15.5,3I4 ,I13,3I1///)
  430 FORMAT(/6H0 VSH=,E12.5,6H  VSM= E12.5,6H  VST=,E12.5,6H  VPH=,E12.
     15,6H  VPM=,E12.5,6H  VPT=,E12.5,/45H0 IK    LITTLE A      LITTLE
     2B          ALPHA    /)
  432 FORMAT (I4,3E15.5)
  434 FORMAT (31H0 IK         M(IK)            N(IK)   /)
  436 FORMAT (I4,2E15.5)
  438 FORMAT (6H0  PH=,E12.5,6H   PM= E12.5,6H   PT=,E12.5,6H   QH=,E12.
     15,6H   QM=,E12.5,6H   QT=,E12.5)
  440 FORMAT (//9H0  WSH = ,E12.5,9H   WMH = ,E12.5,9H   WPH = ,E12.5,/9
     1H   WSM = ,E12.5,9H   WMM = ,E12.5,9H   WPM = ,E12.5,/9H   WST = ,
     2E12.5,9H   WMT = ,E12.5,9H   WPT = ,E12.5//)
  442 FORMAT (8H  A(IK)=,9E12.5)
  444 FORMAT (7H0  N = ,I1,16H  X = WMM/WCR = ,E12.5,18H   WT FLOW CALC
     1= ,E12.5,7H   D = ,E12.5)
  006 READ(5,400)
  008 READ  (5,401)
  010 READ  (5,402)JX,JY,JZ,KR4,KR5,KR6,KR7
  012 IF(KR4) 25,14,25
  014 IF(KR5) 30,16,30
  016 IF(KR6) 18,20,18
  018 IF(KR6-1) 42,34,42
  020 IF(KR7) 22,66,22
  022 IF(KR7-1) 24,48,24
  024 IF(KR7-3) 64,56,64
  025 IF(KR4-9)26,6,26
  026 READ  (5,404)GAMMA,OMEGA,WGIV
  028 GO TO 14
  030 READ  (5,406)RCH,RCT,RH,RT
  032 GO TO 16
```

```
034 READ  (5,408)ENO(1),CS(1),CP(1),WCR(1),PWCR(1)
036 DO 38 K=1,2
    ENO(K+1)=ENO(K)
    CS (K+1)=CS (K)
    CP(K+1)=CP(K)
    WCR(K+1)=WCR(K)
038 PWCR(K+1)=PWCR(K)
040 GO TO 20
042 DO 44 I=1,3
044 READ  (5,408)ENO(I),CS(I),CP(I),WCR(I),PWCR(I)
046 GO TO 20
048 READ  (5,404)BETA(1)
050 DO 52 IK=1,8
052 BETA(IK+1)=BETA(IK)
054 GO TO 66
 56 READ  (5,404)BETA(1),BETA(5),BETA(9)
 58 DO 60 K=1,5,4
    TEMP=(BETA(K+4)-BETA(K))/4.0
 59 DO 60  J= 1,3
    KJ=K+J
    BETA(KJ)=BETA(KJ-1)+TEMP
 60 CONTINUE
062 GO TO 66
 64 READ  (5,404)(BETA(IK),IK=1,9)
066 READ  (5,410)ID1,ID2,X
068 WRITE (6,400)
070 WRITE (6,401)
072 WRITE (6,412)GAMMA,OMEGA,WGIV
074 WRITE (6,414)
076 WRITE (6,416)ENO(1),CS(1),CP(1),WCR(1),PWCR(1),RCH,RH
078 WRITE (6,417)ENO(2),CS(2),CP(2),WCR(2),PWCR(2)
080 WRITE (6,418)ENO(3),CS(3),CP(3),WCR(3),PWCR(3),RCT,RT
082 WRITE (6,420)
084 WRITE (6,416)BETA(1),BETA(2),BETA(3)
086 WRITE (6,417)BETA(4),BETA(5),BETA(6)
088 WRITE (6,418)BETA(7),BETA(8),BETA(9)
090 WRITE (6,422)ID1,ID2,X,JX,JY,JZ,KR4,KR5,KR6,KR7
100 PM     = 0.0
    QM     = 1.0
    MPSET  = 1
    NC     = 1
    KSTAR   = 1
    N      = 0
    XOLD   = 0.0
    TWOOM  = 2.0*OMEGA
    C1     = GAMMA-1.0
    C2     = GAMMA/C1
    C3     = C1/(GAMMA+1.0)
    C4     = 1.0/GAMMA
    C5     = 1.0/C1
    DR     = RT-RH
    DR6    = DR/6.0
    DR16    = DR/16.0
    DR24   = DR/24.0
    DRC     = RCT-RCH
    P=0.0
```

```
101 DO 104 I=1,3
    CALL VSUBX (ENO(I),CS(I),CP(I),P,JX,VX)
104 VS(I) = VX
106 P = 1.0
108 DO 110 I = 1,3
    CALL VSUBX (ENO(I),CS(I),CP(I),P,JX,VX)
110 VP(I) = VX
112 WRITE (6,430)VS(1),VS(2),VS(3),VP(1),VP(2),VP(3)
114 P = 0.0
116 DO 120 IK=1,9
    RK     =RH+P*DR
    RCK    =RCH+P*DRC
    SNB    =SIN(BETA(IK)*.01745329)
    B(IK)  =TWOOM*SNB
    SNB2   =SNB*SNB
    CSB2   =1.0-SNB2
    AK     =(CSB2*RCK)-(SNB2/RK)
    ALF(IK)=AK*DR16
118 WRITE (6,432)IK,AK,B(IK),ALF(IK)
120 P = P+.125
122 EM(4) =        -ALF(5)-ALF(4)
    EM(3) = EM(4)-ALF(4)-ALF(3)
    EM(2) = EM(3)-ALF(3)-ALF(2)
    EM(1) = EM(2)-ALF(2)-ALF(1)
    EM(6) =        +ALF(5)+ALF(6)
    EM(7) = EM(6)+ALF(6)+ALF(7)
    EM(8) = EM(7)+ALF(7)+ALF(8)
    EM(9) = EM(8)+ALF(8)+ALF(9)
    EM(5) =0.0
123 WRITE (6,434)
124 DO 126 IK =1,9
    CAPN(IK)=B(IK)/EXP(EM(IK))
126 WRITE (6,436)IK,EM(IK),CAPN(IK)
128 PH=-DR24*(CAPN(1)+CAPN(5)+2.0*CAPN(3)+4.0*(CAPN(2)+CAPN(4)))
    PT=+DR24*(CAPN(5)+CAPN(9)+2.0*CAPN(7)+4.0*(CAPN(6)+CAPN(8)))
    QH  =EXP(EM(1))
    QT  =EXP(EM(9))
130 WRITE (6,438)PH,PM,PT,QH,QM,QT
140 WM(1) =QH*(X*WCR(2)-PH)/WCR(1)
    WM(3) =QT*(X*WCR(2)-PT)/WCR(3)
    WM(2)=X
    WSH   =WM(1)*VS(1)
    WST =WM(3)*VS(3)
    WPH =WM(1)*VP(1)
    WPT =WM(3)*VP(3)
    WSM  =X*VS(2)
    WPM  =X*VP(2)
142 WRITE (6,440)WSH,WM(1),WPH,WSM,WM(2),WPM,WST,WM(3),WPT
144 DO 166 I=1,3
146 SIGA  =0.0
    P=0.0
    ENOC  =ENO(I)
148 GO TO (150,150,158),JY
150 CPC   =CP(I)
    CSC   =CS(I)
    WMC   =WM(I)
```

```
 152 DO 154 IK=1,9
     CALL VSUBX(ENOC,CSC,CPC,P,JX,VX)
     ZK    =VX*WMC
      TEMPO = 1.-C3*ZK*ZK
      IF(TEMPO)1153,1153,153
1153 TEMPO  = ABS(TEMPO)
      WRITE(6,1234)
1234 FORMAT(56H CALCULATION OF A(IK) AT ST. NO. 152+5 IN CTTD MP ASKS
12341,60HFOR LOG OF NEG. REMAINING CALCULATIONS THIS ITERATION THERE ,
12342 13HFORE INVALID.          )
 153 A(IK) = (TEMPO**C5)*ZK
     SIGA  =SIGA+A(IK)
 154 P  =P+.125
 156 GO TO 164
 158 ELS    =(1.0-C3*WS(I)*WS(I))**C2
     ELP    =(1.0-C3*WP(I)*WP(I))**C2
 160 DO162 IK=1,9
     CAPJ   =ELS+P*(ELP-ELS)
     TEMP   =CAPJ**C4
     A(IK)  =TEMP*SQRT((1.0-CAPJ/TEMP)/C3)
     SIGA =SIGA+A(IK)
 162 P  =P+.125
 164 WRITE (6,442)(A(IK),IK=1,9)
     TMP    =.5*(A(1)+A(9))
 166 S(I) =(SIGA-TMP)*(ENOC/8.0)
 168 N      =N+1
     W      =DR6*(S(1)*PWCR(1)+4.0*S(2)*PWCR(2)+S(3)*PWCR(3))
     D      =1.0-(W/WGIV)
 170 WRITE (6,444)N,X,W,D
 172 GO TO ( 10,174,174,174),JZ
 174 IF (NLIM-N) 176,176,180
 176 WRITE (6,500)NLIM
 500 FORMAT (13H0    NLIMIT (,I3,18H) HAS BEEN REACHED  )
 178 GO TO 10
 180 GO TO (182,186,198,198),JZ
 182 WRITE (6,501)
 501 FORMAT (19H ERROR STOP AT 180    )
 184 GO TO 10
 186 GO TO (188,192,256),KSTAR
 188 KSTAR =2
 190 GO TO 210
 192 KSTAR =3
 194 GO TO 240
 198 IF(ABS(D)-TLIM) 10,200,200
 200 GO TO (202,230,242,256),NC
 202 NC=2
 204 IF(ABS(D)-.05)206,210,210
 206 DX =.05
 208 GO TO 212
 210 DX =.20
 212 X1=X
     W1=W
     D1=D
 214 IF(D.GT.0.0) X=X+DX
 216 IF(D.LT.0.0) X=X-DX
 226 GO TO 140
```

```
230 NC=3
232 IF(D*D1)238,234,240
234 WRITE (6,503)
503 FORMAT (19H ERROR STOP AT 232    )
236 GO TO 10
238 DX =.5*DX
    MPSET =2
240 X2=X
    W2=W
    D2=D
241 GO TO 214
242 IF(D*D2)248,244,250
244 WRITE (6,504)
504 FORMAT (19H ERROR STOP AT 242    )
246 GO TO 10
248 MPSET =2
250 GO TO (252,256),MPSET
252 X1=X2
    W1=W2
    D1=D2
    X2=X
    W2=W
    D2=D
254 GO TO 214
256 X3=X
    W3=W
    D3=D
    CALL PABC(X1,X2,X3,W1,W2,W3,APAB,BPAB,CPAB)
260 GO TO (262,266,290,290),JZ
262 WRITE (6,505)
505 FORMAT (19H ERROR STOP AT 260    )
264 GO TO 10
266 X =-BPAB/(APAB+APAB)
268 IF( .001-ABS(X-XOLD))270,270,10
270 XOLD=X
272 IF(W1-W2) 274,280,280
274 IF(W1-W3)276,140,140
276 X1=X3
    W1=W3
    D1=D3
278 GO TO 140
280 IF(W2-W3)282,140,140
282 X2=X3
    W2=W3
    D2=D3
284 GO TO 140
290 DISC=BPAB**2-4.0*APAB*(CPAB-WGIV)
292 IF (DISC)294,298,298
294 WRITE (6,506)
506 FORMAT(55H PABC FIT GIVES NEGATIVE DISCRIMINANT. PROBABLY CHOKED.)
296 GO TO 10
298 XPL=(-BPAB+SQRT(DISC))/(APAB+APAB)
    XMI=(-BPAB-SQRT(DISC))/(APAB+APAB)
300 GO TO (302,302,306,308),JZ
302 WRITE (6,507)
507 FORMAT (19H ERROR STOP AT 300    )
```

```
304 GO TO 10
306 X = AMIN1(XPL,XMI)
     GO TO 316
308 X = AMAX1(XPL,XMI)
316 IF (ABS(D1)-ABS(D2))318,320,320
318 IF (ABS(D2)-ABS(D3))328,322,322
320 IF (ABS(D1)-ABS(D3))328,326,326
322 X2=X3
    W2=W3
    D2=D3
324 GO TO 328
326 X1=X3
    W1=W3
    D1=D3
328 NC=4
330 GO TO 140
    END
```

```
C              SUBROUTINE PABC
      SUBROUTINE PABC (X1,X2,X3,W1,W2,W3,A,B,C)
      C9 = X1+X2
      C11=X1*X1
      C15=X3-X1
      C18= (W2-W1)/(X2-X1)
      A=(C15*C18-W3+W1)/(C15*C9-X3*X3+C11)
      B=C18-C9*A
      C=W1-X1*B-C11*A
      RETURN
      END
```

```
C              SUBROUTINE VSUBX
      SUBROUTINE VSUB X (EN,CS,CP,P,JX,VX)
  700 GO TO  (702,706),JX
  702 VX=EXP(EN*(CS*(.5-P)+.5*(CS-CP)*(P*P-.25)))
  704 GO TO 714
  706 IF (CS-CP) 712,708,712
  708 VX=EXP(EN*CS*(.5-P))
  710 GO TO 714
  712 VX= (2.0*((P*(CS-CP)+CP)/(CS+CP)))**(EN*CS*CP/(CP-CS))
  714 RETURN
      END
```

List of Program Variables in Main Program

A	array for nine values of $\frac{\rho W}{\rho'' W_{cr}}$
AK	a
ALF	array for nine values of $a \frac{(r_t - r_h)}{16}$
APAB	coefficient of x^2 calculated by subroutine PABC
B	array for nine values of b
BETA	array, input, β
BPAB	coefficient of x calculated by subroutine PABC
C_1	$\gamma - 1$
C_2	$\frac{\gamma}{\gamma - 1}$
C_3	$\frac{\gamma - 1}{\gamma + 1}$
C_4	$\frac{1}{\gamma}$
C_5	$\frac{1}{\gamma - 1}$
CAPJ	$\frac{p}{p''}$
CAPN	array, be^{-EM}
CP	array, input, C_p
CPAB	constant coefficient calculated by PABC
CPC	temporary storage for current value of CP
CS	array, input, C_s
CSB2	$\cos^2\beta$
CSC	temporary storage for current value of CS
D	$1 - \frac{w_{calc}}{w_{giv}}$

D1,D2,D3 temporary storage for various values of D

DISC temporary storage

DR $r_t - r_h$

DR16 $(r_t - r_h)/16$

DR24 $(r_t - r_h)/24$

DR6 $(r_t - r_h)/6$

DRC $(1/r_c)_t - (1/r_c)_h$

DX arbitrary change in x $\left(\text{value of } \left(\frac{W}{W_{cr}}\right)_{mid,m}\right)$ for next calculation of weight flow

ELP $\frac{p_p}{p''}$

ELS $\frac{p_s}{p''}$

EM array, $\int_{r_m}^{r} a\, d\zeta$ (Trapezoidal rule is used)

ENO array, input, n_0

ENOC temporary storage for current value of ENO

GAMMA input, γ

I index of most loops executed 3 times (for hub, mean, tip)

ID1 input, z station code, 4-digit number supplied as input and printed out for identification

ID2 input, z station code, 4-digit number supplied as input and printed out for identification

IK index of most loops executed 9 times

J index of an input storage loop

JX input

JY input

JZ input

K	index for 2 loops controlling storage of input
KJ	temporary storage K + J
KR4,KR5,KR6,KR7	input, read control switches
KSTAR	switch-controls branching for successive values of x, if choke solution is wanted. Initialized by program and automatically stepped.
MPSET	switch-controls branch to continue calculations or call subroutine PABC. Initialized by program and altered as a result of calculations.
N	counter for number of times weight flow is calculated. When N = NLIM calculation for that page of input is stopped and a message written at end of output.
NC	switch-controls branching for successive values of x when subsonic or supersonic solution is wanted
NLIM	constant (supplied in DATA statement) limiting number of times weight flow calculation can occur for 1 sheet of input data
OMEGA	input, ω
P	$\frac{n}{n_0}$
PH	$\int_{r_m}^{r_h} be^{-EM}\, dr$ (Simpson's rule is used)
PM	0
PT	$\int_{r_m}^{r_t} be^{-EM}\, dr$ (Simpson's rule is used)
PWCR	array, input, $\rho''W_{cr}$
QH	e^{EM} at hub
QM	1.0
QT	e^{EM} at tip
RCH	input, $(1/r_c)_h$
RCK	current value of $1/r_c$

RCT	input, $(1/r_c)_t$
RH	input, r_h
RK	current value of r
RT	input, r_t
S	array, values of $\int_0^{n_0} \frac{\rho W}{\rho'' W_{cr}} dn$ at hub, mean, and tip
SIGA	temporary storage, $\sum A(IK)$
SNB2	$\sin^2\beta$
SNB	$\sin \beta$
TEMP	temporary storage
TEMPO	temporary storage
TLIM	constant (supplied in DATA statement). Calculation is terminated if TLIM $> \left\| 1 - \frac{w_{calc}}{w_{giv}} \right\|$ whenever subsonic or supersonic solution is sought.
TMP	temporary storage
TWOOM	2ω
VP	array, three values of velocity (calculated in subroutine VSUBX) for pressure surface at hub, mean, and tip
VS	array, three values of velocity (calculated in subroutine VSUBX) for suction surface at hub, mean, and tip
VX	ratio of velocity to mid-channel velocity
W	weight flow calculated for current value of $\left(\frac{W}{W_{cr}}\right)_{mid,m}$
W1,W2,W3	temporary storage for various values of weight flow
WCR	array, input, W_{cr}
WGIV	input, w_{giv}
WM	array, mid-channel values of $\frac{W}{W_{cr}}$ at hub, mean, and tip

WMC	current value of WM
WP	array, $\frac{W}{W_{cr}}$ on pressure surface at hub, mean, and tip
WPH	WP(1)
WPM	WP(2)
WPT	WP(3)
WS	array, $\frac{W}{W_{cr}}$ on suction surface at hub, mean, and tip
WSH	WS(1)
WSM	WS(2)
WST	WS(3)
X	input, also the current value of $\left(\frac{W}{W_{cr}}\right)_{mid,m}$ for which the weight flow is calculated
X1,X2,X3	temporary storage for various values of x
XMI	the lesser of the two values of x at which the line $y = w_{giv}$ intersects the parabola $y = APAB(x^2) + BPAB(x) + CPAB$
XOLD	temporary storage for a value of x when choking weight flow is sought
XPL	the larger of the two values of x at which the line $y = w_{giv}$ intersects the parabola $y = APAB(x^2) + BPAB(x) + CPAB$
ZK	current value of $\frac{W}{W_{cr}}$ at any point

APPENDIX - ANALYTICAL EQUATIONS

In reference 1 a differential equation is derived for the variation of velocities in a radial direction based on the assumption of a mid-channel stream surface (axial symmetry) which is radial at any fixed axial position (i.e., $\frac{\partial z}{\partial r} = 0$ on the midchannel stream surface). This equation is

$$\left.\begin{aligned} &\frac{dW}{dn} = aW - b \\ &\text{where} \\ &a = \frac{\cos^2\beta}{r_c} - \frac{\sin^2\beta}{r\cos\alpha} \\ &b = \sin\beta\left(\frac{2\omega}{\cos\alpha} + \tan\alpha\,\frac{dW_\theta}{dm}\right) \end{aligned}\right\} \tag{A1}$$

The coordinate system is shown in figure 6.

With the assumption that α is sufficiently small so that $\cos\alpha$ is nearly 1 and $\tan\alpha$ is nearly zero, we can set $dn = dr$ to obtain

$$\left.\begin{aligned} &\frac{dW}{dr} = aW - b \\ &\text{where} \\ &a = \frac{\cos^2\beta}{r_c} - \frac{\sin^2\beta}{r} \\ &b = 2\omega\sin\beta \end{aligned}\right\} \tag{A2}$$

An analytical solution of equation (A2) is

$$W(r) = e^{\int_{r_m}^{r} a(\zeta)d\zeta}\left(W_m - \int_{r_m}^{r} b(\xi)e^{-\int_{r_m}^{\xi} a(\zeta)d\zeta}\,d\xi\right) \tag{A3}$$

for the midchannel stream sheet.

$W_{mid,m}$ is computed from the input values $W/W_{cr}{}_{mid,m}$ 44 and $W_{cr,m}$ 27. With equation (A3) the midchannel velocities at the hub and tip can then be calculated from the value of $W_{mid,m}$. For computing the integration in equation (A3), the interval r_h to r_t is divided into eight equal intervals. Since the integration is started at the mean radius, only four integration steps are required in each direction. The integration is done numerically using the trapezoidal rule. In the case where β is given as input only at hub, mean, and tip (rather than at all nine stations) linear interpolation is used to determine the values which are not given. The meridional streamline curvature, $\frac{1}{r_c}$, is assumed to vary linearly from hub to tip.

The ratios of surface velocity to midchannel velocity, $\left(\frac{W_s}{W_{mid}}\right.$ and $\left.\frac{W_p}{W_{mid}}\right)$ can be computed based on an assumption of linear variation of either streamline curvature or streamline radius of curvature. These equations are derived in the next section. If linear variation of curvature is assumed,

$$\frac{W}{W_{mid}} = e^{n_0\left[C_s\left(\frac{3}{8} - \frac{n}{n_0}\right) + \frac{1}{8} C_p - \frac{1}{2} (C_p - C_s)\left(\frac{n}{n_0}\right)^2\right]} \tag{A4}$$

where n is the distance along the orthogonal from the blade suction surface. Thus, the velocity can be determined at any point on the orthogonal from (A4). If, on the other hand, linear variation of radius of curvature is assumed

$$\frac{W}{W_{mid}} = \begin{cases} \left(\dfrac{2\left(C_p + (C_s - C_p)\frac{n}{n_0}\right)}{C_p + C_s}\right)^{\frac{n_0 C_s C_p}{C_p - C_s}} & \text{if } C_p \neq C_s \\ e^{n_0\left(\frac{1}{2} - \frac{n}{n_0}\right)C_s} & \text{if } C_p = C_s \end{cases} \tag{A5}$$

By substituting $n = 0$, and $n = n_0$ in equation (A4) or equation (A5), and using the results from equation (A3), the blade surface velocity at hub, mean and tip can be computed.

With the velocities obtained, the weight flow past the orthogonal can be computed. This is calculated from

$$w_{calc} = \int_{r_h}^{r_t} \int_0^{n_0} \rho W \, dn \, dr \tag{A6}$$

The inner integral is computed at hub, mean, and tip using eight equal intervals from suction to pressure surface, and using the trapezoidal rule. If JY = 1 or 2 (linear curvature or radius of curvature variation) the critical velocity ratios, $\frac{W}{W_{cr}}$, are computed using the input value of W_{cr} and equations (A4) or (A5), and (A3). The weight flow parameter, $\frac{\rho W}{\rho'' W_{cr}}$, is then calculated from

$$\frac{\rho W}{\rho'' W_{cr}} = \left[1 - \frac{\gamma - 1}{\gamma + 1}\left(\frac{W}{W_{cr}}\right)^2\right]^{\frac{1}{\gamma - 1}} \frac{W}{W_{cr}} \tag{A7}$$

If JY = 3 (linear variation of static pressure), then the blade surface static pressures are calculated based on the surface velocities

$$\frac{p}{p''} = \left[1 - \frac{\gamma - 1}{\gamma + 1}\left(\frac{W}{W_{cr}}\right)^2\right]^{\frac{\gamma}{\gamma - 1}} \tag{A8}$$

and then pressures along the orthogonal are calculated by

$$\frac{p}{p''} = \frac{p_s}{p''} + \frac{n}{n_0}\left(\frac{p_p}{p''} - \frac{p_s}{p''}\right) \tag{A9}$$

then

$$\frac{\rho W}{\rho'' W_{cr}} = \left(\frac{p}{p''}\right)^{\frac{1}{\gamma}} \left\{\frac{\gamma + 1}{\gamma - 1}\left[1 - \left(\frac{p}{p''}\right)^{\frac{\gamma - 1}{\gamma}}\right]\right\}^{1/2} \tag{A10}$$

After computing the inner integral in equation (A6) at hub, mean, and tip, the outer integral is approximated by Simpson's Rule. If JZ = 1, no further calculations are made.

If JZ = 2, further calculations are made with new values of $\left(\frac{W}{W_{cr}}\right)_{mid,m}$ to get the weight flow, w, as a function of $\left(\frac{W}{W_{cr}}\right)_{mid,m}$.

Using a parabolic approximation, the value of $\left(\frac{W}{W_{cr}}\right)_{mid,m}$ giving maximum weight flow can be estimated. When two successive estimates of $\left(\frac{W}{W_{cr}}\right)_{mid,m}$ differ by less than 0.001, the computation is stopped. The maximum weight flow is normally determined within 5 iterations.

If JZ = 3 or 4 further calculations are made to determine a value of $(W/W_{cr})_{mid,m}$ which will give a weight flow, w, which is close to the input value, w_{giv}. If w_{giv} is less than choking weight flow, there are two values of $(W/W_{cr})_{mid,m}$ which will give $w_{calc} = w_{giv}$. If JZ = 3 the smaller, or subsonic, value of $(W/W_{cr})_{mid,m}$ will be found. If JZ = 4 the larger, or supersonic, solution will be found. When $\left| w_{giv} - w_{calc} \right| < 0.001\ w_{giv}$ the calculations are stopped. A solution is normally found within five iterations. If w_{giv} is larger than the choking weight flow no solution exists. In this case calculations are made for five values of $(W/W_{cr})_{mid,m}$ and the choking weight flow can usually be estimated from these values.

Blade to Blade Velocity Variation

The method of calculating blade surface velocities from mid-channel velocities is based on reference 2. The assumptions that the flow is steady relative to the blade, nonviscous, and isentropic along the blade to blade streamline orthogonal yield

$$\frac{dW}{dn} = - \frac{W}{r_c} \tag{A11}$$

This is equation (1) of reference 2 and can be derived from the force equation.

If it is assumed that the curvature, $C = 1/r_c$, varies linearly along the orthogonal, the equation can be integrated to obtain

$$\frac{W}{W_{mid}} = e^{-\frac{n_0}{2(C_p - C_s)}\left[C^2 - \left(\frac{C_p + C_s}{2}\right)^2\right]} \tag{A12}$$

which is equation (7) of reference 2. Since the curvature is assumed to vary linearly,

$$C = C_s + (C_p - C_s)\frac{n}{n_0} \tag{A13}$$

When equation (A13) is substituted in equation (A12), equation (A4) is obtained.

If it is assumed that the radius of curvature varies linearly, then

$$r_c = (r_c)_s + \left[(r_c)_p - (r_c)_s\right]\frac{n}{n_0} \tag{A14}$$

Using this in equation (A11) we have

$$\frac{dW}{dn} = - \frac{W}{(r_c)_s + \left[(r_c)_p - (r_c)_s\right]\frac{n}{n_0}}$$

Integrating equation (A15) from the mid-channel gives

$$\log \frac{W}{W_{mid}} = \frac{n}{(r_c)_p - (r_c)_s} \log\left[\frac{(r_c)_s + \frac{\left[(r_c)_p - (r_c)_s\right]}{2}}{(r_c)_s + \left[(r_c)_p - (r_c)_s\right]\frac{n}{n_0}}\right] \tag{A16}$$

Substituting $r_c = \frac{1}{C}$ into equation (A16) and solving for $\frac{W}{W_{mid}}$, equation (A5) is obtained when $C_s \neq C_p$. For the special case where $C_s = C_p$, r_c is a constant in equation (A11), and an integration gives

$$\log \frac{W}{W_{mid}} = - \frac{\left(n - \frac{n_0}{2}\right)}{r_c} = C_s n_0\left(\frac{1}{2} - \frac{n}{n_0}\right) \tag{A17}$$

which is equivalent to equation (A5) when $C_s = C_p$.

References

1. Hamrick, Joseph T.; Ginsburg, Ambrose; and Osborn, Walter M.: Method of Analysis for Compressible Flow Through Mixed-Flow Centrifugal Impellers of Arbitrary Design. NACA TR 1082, 1952.

2. Huppert, M. C.; and MacGregor, Charles: Comparison Between Predicted and Observed Performance of Gas-Turbine Stator Blade Designed for Free-Vortex Flow. NACA TN 1810, 1949.

3. Stewart, Warner L.; Wong, Robert Y.; and Evans, David G.: Design and Experimental Investigation of Transonic Turbine With Slight Negative Reaction Across Rotor Hub. NACA RM E53L29a, 1954.

4. Stewart, Warner L.; Whitney, Warren J.; and Schum, Harold J.: Three-Dimensional Flow Considerations in the Design of Turbines. Paper No. 59-Hyd-1, ASME, 1959.

5. Shepherd, D. G.: Principles of Turbomachinery. The MacMillan Co., 1956.

6. Miser, James W.; Stewart, Warner L.; and Whitney, Warren J.: Analysis of Turbomachine Viscous Losses Affected by Changes in Blade Geometry. NACA RM E56F21, 1956.

TABLE I

```
1 SAMPLE PROBLEM

4 NEGATIVE REACTION TURBINE

 12 GAMMA =  0.14000E 01 13 OMEGA =  0.48800E 03 14 WGIVEN =  0.59170E 00

             N ZERO            C SUB S          C SUB P          W SUB CR         RHO W CR      1/R SUB C          R

   HUB     19 0.42080E-01  20 0.15710E 02  21 0.88200E 01  22 0.94970E 03  23 0.47160E 02  15 0.            17 0.87500E 00
   MEAN    24 0.54500E-01  25 0.14160E 02  26 0.84200E 01  27 0.95740E 03  28 0.49520E 02
   TIP     29 0.71500E-01  30 0.12000E 02  31 0.63000E 01  32 0.96660E 03  33 0.52540E 02  16 0.            18 0.12500E 01

             BETA

   HUB     34 -0.50000E 01    -0.78250E 01    -0.10650E 02
   MEAN       -0.13475E 02 35 -0.16300E 02    -0.19350E 02
   TIP        -0.22400E 02    -0.25450E 02 36 -0.28500E 02

  1000    3     0.80000E 00   1    1    3           1133
      43            44        5    6    7        8  9  10  11

 VSH= 0.13422E 01  VSM= 0.14145E 01  VST= 0.14594E 01  VPH= 0.80106E 00  VPM= 0.76449E 00  VPT= 0.75868E 00

 IK    LITTLE A       LITTLE B        ALPHA
  1   -0.86813E-02   -0.85064E 02   -0.20347E-03
  2   -0.20107E-01   -0.13288E 03   -0.47126E-03
  3   -0.35256E-01   -0.18037E 03   -0.82632E-03
  4   -0.53463E-01   -0.22743E 03   -0.12530E-02
  5   -0.74140E-01   -0.27393E 03   -0.17377E-02
  6   -0.98961E-01   -0.32339E 03   -0.23194E-02
  7   -0.12559E 00   -0.37192E 03   -0.29435E-02
  8   -0.15349E 00   -0.41941E 03   -0.35973E-02
  9   -0.18214E 00   -0.46571E 03   -0.42690E-02

 IK      M(IK)           N(IK)
  1    0.70424E-02   -0.84467E 02
  2    0.63676E-02   -0.13204E 03
  3    0.50701E-02   -0.17946E 03
  4    0.29907E-02   -0.22675E 03
  5    0.            -0.27393E 03
  6   -0.40571E-02   -0.32470E 03
  7   -0.93200E-02   -0.37541E 03
  8   -0.15861E-01   -0.42612E 03
  9   -0.23727E-01   -0.47689E 03

  PH= 0.33632E 02   PM= 0.            PT=-0.70389E 02   QH= 0.10071E 01   QM= 0.10000E 01   QT= 0.97655E 00


  WSH =   0.10422E 01   WMH =   0.77652E 00   WPH =   0.62204E 00
  WSM =   0.11316E 01   WMM =   0.80000E 00   WPM =   0.61159E 00
  WST =   0.12331E 01   WMT =   0.84492E 00   WPT =   0.64103E 00

 A(IK)= 0.63259E 00 0.63282E 00 0.62495E 00 0.61191E 00 0.59588E 00 0.57841E 00 0.56057E 00 0.54312E 00 0.52655E 00
 A(IK)= 0.62095E 00 0.63325E 00 0.63139E 00 0.62024E 00 0.60343E 00 0.58355E 00 0.56239E 00 0.54117F 00 0.52068E 00
 A(IK)= 0.59386E 00 0.62463E 00 0.63389E 00 0.62903E 00 0.61557E 00 0.59740E 00 0.57717E 00 0.55661E 00 0.53685E 00

  N = 1   X = WMM/WCR =   0.80000E 00   WT FLOW CALC =   0.61444E 00    D = -0.38439E-01


  WSH =   0.97410E 00   WMH =   0.72576E 00   WPH =   0.58138E 00
  WSM =   0.10609E 01   WMM =   0.75000E 00   WPM =   0.57337E 00
  WST =   0.11625E 01   WMT =   0.79656E 00   WPT =   0.60434E 00

 A(IK)= 0.63343E 00 0.62613E 00 0.61270E 00 0.59566E 00 0.57681E 00 0.55739E 00 0.53826E 00 0.51999E 00 0.50293E 00
 A(IK)= 0.63114E 00 0.63304E 00 0.62354E 00 0.60695E 00 0.58638E 00 0.56399E 00 0.54122E 00 0.51904E 00 0.49802E 00
 A(IK)= 0.61421E 00 0.63222E 00 0.63204E 00 0.62046E 00 0.60237E 00 0.58110E 00 0.55885E 00 0.53703E 00 0.51653E 00

  N = 2   X = WMM/WCR =   0.75000E 00   WT FLOW CALC =   0.60127E 00    D = -0.16173E-01


  WSH =   0.90597E 00   WMH =   0.67500E 00   WPH =   0.54072E 00
  WSM =   0.99013E 00   WMM =   0.70000E 00   WPM =   0.53514E 00
  WST =   0.10919E 01   WMT =   0.74819E 00   WPT =   0.56764E 00

 A(IK)= 0.62719E 00 0.61340E 00 0.59535E 00 0.57509E 00 0.55406E 00 0.53323E 00 0.51324E 00 0.49450E 00 0.47723E 00
 A(IK)= 0.63386E 00 0.62649E 00 0.61040E 00 0.58929E 00 0.56572E 00 0.54142E 00 0.51754E 00 0.49477F 00 0.47355F 00
 A(IK)= 0.62756E 00 0.63374E 00 0.62510E 00 0.60769E 00 0.58570E 00 0.56192E 00 0.53812E 00 0.51543E 00 0.49447E 00

  N = 3   X = WMM/WCR =   0.70000E 00   WT FLOW CALC =   0.58401E 00    D =  0.12990E-01


  WSH =   0.93436E 00   WMH =   0.69615E 00   WPH =   0.55766E 00
  WSM =   0.10196E 01   WMM =   0.72083E 00   WPM =   0.55107E 00
  WST =   0.11214E 01   WMT =   0.76835E 00   WPT =   0.58293E 00

 A(IK)= 0.63065E 00 0.61943E 00 0.60319E 00 0.58418E 00 0.56398E 00 0.54367E 00 0.52399E 00 0.50540E 00 0.48819E 00
 A(IK)= 0.63365E 00 0.62999E 00 0.61651E 00 0.59717E 00 0.57476E 00 0.55119E 00 0.52771E 00 0.50514E 00 0.48397E 00
 A(IK)= 0.62287E 00 0.63385E 00 0.62861E 00 0.61352E 00 0.59306E 00 0.57025E 00 0.54705E 00 0.52467E 00 0.50387E 00

  N = 4   X = WMM/WCR =   0.72083E 00   WT FLOW CALC =   0.59170E 00    D =  0.60722E-05
```

E-3580

E-3580



Figure 1. - Pair of typical turbine blades with three-dimensional orthogonal surface across flow passage.



Figure 2. - Blade channel - mean section.



Figure 3. - Blade channel - mean section. (Showing typical streamline orthogonal.)



Figure 4. - Meridional plane wall curvature.

E-3580

cttd$_{FIV}$ INPUT DATA SHEET

The lowest line in each block displays Roman numerals for card sequence and arabic numerals for card columns within each card.

[All numerical values (columns 12 → 42, 44) are entered in nine-column fields, (read in with a FORTRAN format specification of E9.5) usually as ±xxxxx±zz representing ±.xxxxx multiplied by 10$^{\pm zz}$.]

		1
	1	*Sample problem*
C.C.	I 1	I 2-40(MAY BE ALPHANUMERIC)

		4
	0	*Negative reaction turbine*
C.C.	II 1	II 2-40(MAY BE ALPHANUMERIC)

	5	6	7	8	9	10	11	12	13	14	15	16	17	18
									1/sec	lb/sec	1/ft	1/ft	ft	ft
	JX	JY	JZ	KR 4	KR 5	KR 6	KR 7	γ	ω	w_{giv}	$(1/r_c)_h$	$(1/r_c)_t$	r_h	r_t
	1	*1*	*3*	*1*	*1*	*3*	*3*	*1.4*	*488.*	*0.5917*	*0*	*0*	*0.875*	*1.25*
C.C.	III 1	2	3	4	5	6	7	IV 1-9	10-18	19-27	V 1-9	10-18	19-27	28-36

	19	20	21	22	23	24	25	26	27	28
	ft	1/ft	1/ft	ft/sec	lb/ft^2 sec	ft	1/ft	1/ft	ft/sec	lb/ft^2 sec
	$n_{0,h}$	$C_{s,h}$	$C_{p,h}$	$W_{cr,h}$	$\rho'' W_{cr,h}$	$n_{0,m}$	$C_{s,m}$	$C_{p,m}$	$W_{cr,m}$	$\rho'' W_{cr,m}$
	0.04208	*15.71*	*8.82*	*949.7*	*47.16*	*0.0545*	*14.16*	*8.42*	*957.4*	*49.52*
C.C.	VI-A 1-9	10-18	19-27	28-36	37-45	VI-B 1-9	10-18	19-27	28-36	37-45

	29	30	31	32	33
	ft	1/ft	1/ft	ft/sec	lb/ft^2 sec
	$n_{0,t}$	$C_{s,t}$	$C_{p,t}$	$W_{cr,t}$	$\rho'' W_{cr,t}$
	.0715	*12.00*	*6.30*	*966.6*	*52.54*
C.C.	VI-C 1-9	10-18	19-27	28-36	37-45

	34	35	36	37	38	39	40	41	42
	deg β								
	-5.0	*-16.3*	*-28.5*	←		*Omit*			→
C.C.	VII-A 1-9	10-18	19-27	VII-B 1-9	10-18	19-27	VII-C 1-9	10-18	19-27

		43	44
		z station code	$\left(\frac{W}{W_{cr}}\right)_{mid,m}$
	0	*10000003*	*0.8*
C.C.	VIII 1	2-9 must be (NUMERIC)	10-18

Figure 5. - Input data sheet.



Figure 6. - Coordinate system and velocity components.